

Appendix A.2: Introduction to Maple

Maple is an application that can be used to solve algebraic and differential equations. Maple has the ability to do mathematics in symbolic form and therefore it can determine the analytical solution to algebraic and differential equations. The capability to differentiate, integrate and algebraically manipulate mathematical expressions in symbolic form can be a very powerful aid in solving heat transfer problems. Also, Maple provides a very convenient mathematical reference; if, for example, you've forgotten that the derivative of sine is cosine, it is easy to use Maple to quickly provide this information. Maple can replace numerous mathematical reference books that might otherwise be required to carry out all of the integration, differentiation, simplification, etc. required to solve many engineering problems.

Maple and EES can be used effectively together; Maple can determine the analytical solution to a problem and these symbolic expressions can subsequently be copied (almost directly) into EES for convenient numerical evaluation and manipulation in the context of a specific application. Although Maple can also do numerical evaluation, unit conversion, parametric studies, optimization and plots, our experience is that it is more convenient to use EES for these tasks. Combining the capabilities of both programs provides the user with a very powerful set of mathematical tools.

Maple has a rich set of commands that provide several different ways to solve a problem. A quick internet search will reveal that many books have been written on how to use Maple effectively. Only a subset of Maple's full capabilities are required to effectively solve heat transfer problems. This appendix will summarize the commands that we have found to be most useful and are used throughout this text.

All of the examples in this book can be solved with little or no modifications using any version of Maple. If you do not have Maple installed on your computer, you can still access Maple on the internet through the Maple Command Applet. All of the capabilities needed to solve the heat transfer problems in this book are provided by the Command Applet. The commands are entered, one at a time, in the Applet window. At the time of this writing, the Command Applet is available at:

<http://maplenet.maplesoft.com/maplenet/tutorial/MapleLockingApplet.html>

Additional information concerning Maple can be obtained from:

MapleSoft, a division of Waterloo Maple Inc.

<http://www.maplesoft.com/>

info@maplesoft.com

A.2.1 Getting Started

After starting Maple, you may be asked what document mode to use and given the choice of Document or Worksheet. Select Worksheet mode to bring up an edit window with a > character and a blinking cursor. Maple commands are entered to the right of the > character.

The Maple output format may not appear on your computer as it does in this text. You can change the output format by selecting Options from the Tools menu and then clicking on the Display Tab. Make sure that the Input display is set to Maple Notation and the Output display is set to 2-D Math Notation.

Some general rules that will be helpful in entering input to Maple are:

1. Assignment of a name or variable to a mathematical object is done using the := operator (take careful note of the colon that appears before the equal sign). Note that the equal sign = designates equality, which yields an expression but not an assignment. The most common mistake that new users will make using Maple is to enter $x = 5$ when they really mean $x := 5$; $x:=5$ assigns the variable x to have a value of 5.
2. Each instruction to Maple must end with a colon (:) or a semi-colon (;). If the instruction ends with a semi-colon, Maple will acknowledge the command by displaying it in its own format. If the colon is used to terminate the line, Maple will accept the instruction with no acknowledgement.
3. Maple input is case-sensitive; x is not the same as X in Maple.
4. Multiplication is indicated with the * character.
5. Raising a variable or number to a power is indicated by the ^ character.
6. The pound sign (#) is used to indicate comments. Everything following the # sign to the end of the line is ignored.
7. Maple provides extensive online help. To obtain help, enter ? followed by the subject for which help is needed, e.g., ?integration. Help can also be obtained using the Help menu.
8. Variables remain assigned to whatever value or expression that they were last assigned until and unless they are reassigned or cleared. To determine the current assignment of a variable, enter its name followed by a semi-colon;
9. The % command can be used a shorthand expression to represent the result of the previous command.
10. Enter **restart**; to clear all previously assigned variables.

In its simplest form, Maple can serve as a calculator. It even provides a **convert** function for unit conversions. For example, to enter a diameter of 3 ft and calculate the area in m^2 , enter the following commands.

```
> d:=3; # diameter in ft
                                     d := 3
> r:=d/2; # radius in ft
                                     r :=  $\frac{3}{2}$ 
> A:=pi*r^2; # area in ft^2; pi is a built-in constant
```

```

A := 9 π / 4
> A_m2:=convert(A, 'units','ft^2','m^2'); # A_m is the area in m^2
A_m2 = 1306449 π / 6250000

```

The `convert` command in Maple is extensive and it provides many other capabilities beyond unit conversion. For example, it is easy to convert the equation

$$T = C_1 \exp(-mx) + C_2 \exp(mx) \quad (\text{A.2.1})$$

into an equivalent form involving hyperbolic sine and cosine using the `convert` command and the flag 'trigh' which indicates hyperbolic trigonometric format:

```

> T:=C1*exp(-m*x)+C2*exp(m*x);
T := C1 e(-mx) + C2 e(mx)
> TH:=convert(T,'trigh');
TH := (C1 + C2) cosh(mx) + (-C1 + C2) sinh(mx)

```

The single quotes around `trigh` are optional. There are many other options for the second parameter of the `convert` function, as can be seen by viewing the Maple online help, but they are not needed for solving the problems in this book.

A.2.2 Evaluating Expressions and Solving Equations

One of the most useful capabilities of Maple is its ability to analytically solve algebraic equations in symbolic form. This capability will be demonstrated by solving a quadratic equation, $y = ax^2 + bx + c$. First, specify the equation y .

```

> y:=a*x^2+b*x+c; # quadratic equation
y := a x2 + b x + c

```

The values of a , b , and c , as well as x , are not yet specified. We can evaluate y at specific values of these parameters using the `eval` command; the first argument of the `eval` command is the expression while the second is the substitution. For example, to obtain a symbolic expression for y evaluated at $x=1$:

```

> eval(y,x=1);
a + b + c

```

The evaluation can occur with multiple substitutions by specifying the values of more than one variable; in this case, the list of specifications must be enclosed in curly braces and separated by commas. The value of y at $x = 1$, $a = -2$, $b = 3$, and $c = 4$ is obtained according to:

```
> y1:=eval(y,{x=1,a=-2,b=3,c=4}); # evaluate y at given values
      y1 := 5
```

In addition to simple evaluations, Maple can solve an equation using the **solve** command. The first argument of the **solve** command is the equation to be solved while the second is the variable that should be solved for. For example, the value(s) of x that satisfies the equation $y = -2$ can be determined using the **solve** command, as shown below. The result of the **solve** command is placed in variable xs . Note that a , b and c have not been assigned to values at this point, so it is necessary to tell Maple which of the unspecified variables we wish to solve for, x in this case, and the solution will be expressed symbolically in terms of the remaining variables.

```
> xs:=solve(y=-2,x); # solve the equation for x
      xs :=  $\frac{-b + \sqrt{b^2 - 4ac - 8a}}{2a}$ ,  $\frac{-b - \sqrt{b^2 - 4ac - 8a}}{2a}$ 
```

There are two solutions to the quadratic equation and Maple has identified both. The variable xs contains both solutions in two elements, $xs[1]$ and $xs[2]$:

```
> xs[1];
       $-\frac{b - \sqrt{b^2 - 4ac - 8a}}{2a}$ 
> xs[2];
>
       $-\frac{b + \sqrt{b^2 - 4ac - 8a}}{2a}$ 
```

We can set values for a , b , and c and then determine the numerical, as opposed to symbolic, solutions to the equation.

```
> a:=-2; b:=3; c:=4;
      a := -2
      b := 3
      c := 4
> x1:=xs[1]; # the first of the two solutions
      x1 :=  $\frac{3}{4} - \frac{\sqrt{57}}{4}$ 
```

Notice that Maple displays results in analytical form when it can; using the **evalf** function results in the value being displayed in floating point format.

```
> x2:=evalf(xs[2]); # the 2nd solution, displayed in floating pt
format
x2 := 2.637458609
```

A.2.3 Substitution and Simplification

The ability to symbolically manipulate equations provided by Maple is extensive. It is further enhanced by the **subs** (substitute) command that can be used to substitute a numerical value or symbolic expression in place of a variable. For example, suppose that you know that

$$T = \frac{ax^2 + bx + c}{1 - x} \quad (\text{A.2.2})$$

and you wish to apply a coordinate transformation for which

$$x = \frac{1}{z + 1} \quad (\text{A.2.3})$$

Enter the equation into Maple and apply the **subs** command for the transformation.

```
> restart;
> T:=(a*x^2+b*x+c)/(1-x);
T := \frac{ax^2 + bx + c}{1 - x}
> Tz:=subs(x=1/(z+1),T);
Tz := \frac{\frac{a}{(z+1)^2} + \frac{b}{z+1} + c}{1 - \frac{1}{z+1}}
```

Note that the result provided by Maple can often be expressed in an equivalent but algebraically simpler manner by using the **simplify** command.

```
> simplify(%);
\frac{a + bz + b + cz^2 + 2cz + c}{(z + 1)z}
```

The **subs** command can also be used to substitute numerical values for a variable. For example

```
> subs(z=1,%);
```

$$\frac{a}{2} + b + 2c$$

The **eval** command would provide the same result:

```
> eval(%, z=1);
```

$$\frac{a}{2} + b + 2c$$

The **subs** and **eval** commands have overlapping capability. Note that % command is shorthand for the previous result from Maple.

A.2.4 Differentiation, Integration and Limits

Maple has the capability to symbolically differentiate a mathematical expression using its **diff** function. Partial differentiation of functions of multiple variables can be accomplished by indicating the variable that the expression is to be differentiated with respect to.

The following example starts by clearing the variable space with the **restart** command. Then a symbolic equation $f = x^2y + 3xy + 4x \ln(y)$ is specified.

```
> restart;
> f:=x^2*y+3*x*y+4*x*ln(y); # symbolic equation
      f := x2 y + 3 x y + 4 x ln(y)
```

The partial derivatives of f with respect to x and y can be found in symbolic form using the **diff** command. The **diff** command takes two arguments; the first is the symbolic expression to be differentiated and the second is the differentiation variable.

```
> dfdx:=diff(f,x); #partial derivative of f with respect to x
      dfdx := 2 x y + 3 y + 4 ln(y)
> dfdy:=diff(f,y); #partial derivative of f with respect to y
      dfdy := x2 + 3 x +  $\frac{4x}{y}$ 
```

The **eval** command can be used to evaluate the differentials at a specific combination of variables. Note that a specification of the values for more than one variable is enclosed in curly braces.

```
> eval(dfdx, x=1);
```

$$5y + 4 \ln(y)$$

```
> eval(dfdx, {x=2, y=1});
```

The **eval** and **diff** commands can be handy when you forget the relationships among the trigonometric functions; you can use Maple as you would a table of integration or differentiation rules in a reference book:

```
> f:=sin(a*x);
                                f:= sin(a x)
> dfdx:=diff(f,x);
                                dfdx := cos(a x) a
```

The **int** command can be used to find the definite or indefinite integral of an expression. Clear the variable space and redefine f according to:

```
> restart;
> f:=x*cos(x)+b/x; #new function
                                f:= x cos(x) +  $\frac{b}{x}$ 
```

The symbolic variable gi is assigned to be the indefinite integral of f with respect x using the **int** command; Maple responds to this input with the integral in symbolic form.

```
> gi:=int(f,x); #indefinite integral of f w/respect to x
                                gi := cos(x) + x sin(x) + b ln(x)
```

The definite integral of f from 1 to 2 can be evaluated symbolically as well.

```
> gd:=int(f,x=1..2); #definite integral from 1 to 2
                                gd := -cos(1) - sin(1) + cos(2) + b ln(2) + 2 sin(2)
```

It is also possible to obtain a symbolic expression for the definite integral that includes the limits of integration (e.g., the integration of f between x_i and x_o):

```
> assume(x_i>0);
> assume(x_o>0);
> gd2:=int(f,x=x_i..x_o);
                                gd2 := -cos(x_i~) - b ln(x_i~) - sin(x_i~) x_i~ + cos(x_o~) + b ln(x_o~)
                                + sin(x_o~) x_o~
```

Note that it was necessary to specify that the limits were both positive with the **assume** command in order to obtain a symbolic expression due to the singularity associated with $x=0$. The **assume** command also allows a variable to be specified to be a real number or an integer.

The ~ is appended to variables for which the **assume** command has been used; this formatting option can be enabled or disabled with the Options menu.

Situations are often encountered in heat transfer problems for which it is necessary to determine limits of symbolic expressions (these arise often in the context of boundary conditions). As an example, suppose that you wish to evaluate y where

$$y = \lim_{x \rightarrow 0} x \ln(x) \quad (\text{A.2.4})$$

Entering the equation into a calculator or EES will not yield a result because $\ln(0)$ is negative infinity. It is, of course, possible to evaluate this limit using L'Hospital's rule, but it is easier to apply the **limit** command in Maple.

```
> limit(x*ln(x), x=0);
0
```

Maple is able to determine the limits of both determinate and indeterminate forms. It is particularly useful in finding limits of functions. For example, the governing equation for many heat transfer problems is a second order differential equation that is often called Bessel's equation. There are a number of different functions (e.g., BesselI, BesselJ, etc.) with different orders that provide solutions to the equation, as explained in Section 1.8. It is often necessary to know the value of the function at the boundaries of the computational domain, e.g., $x = 0$ or $x \rightarrow \infty$. Maple can provide these limiting values.

```
> limit(BesselI(0,x), x=0);
1
> limit(BesselJ(0,x), x=0);
1
> limit(BesselI(0,x), x=infinity);
∞
> limit(BesselJ(0,x), x=infinity);
0
```

A.2.5 Using Maple with EES

Maple and EES can be used together very effectively; the symbolic manipulations (integrations, differentiation, etc.) that are required to solve a problem can be carried out in Maple and the results copied directly into EES where they can be numerically evaluated, parametrically varied, optimized, plotted, etc. The engineer provides the physical insight that is required to generate the mathematical model (typically expressed as a governing differential equation with boundary conditions), Maple is used to symbolically solve the mathematical model, EES is used to work with the solution, and finally the engineer must verify that the results are correct and useful.

In the Section A.2.6 we will see that this process is particularly attractive when solving differential equations that are subject to specific boundary conditions. We will illustrate the process using a very simple example in this section.

We see in Chapter 3 that temperature distribution within a semi-infinite body exposed to a step change in its surface temperature is given by:

$$T = T_i + (T_s - T_i) \left[1 - \operatorname{erf} \left(\frac{x}{2\sqrt{\alpha t}} \right) \right] \quad (\text{A.2.5})$$

where T_s and T_i are the surface and initial temperatures of the body, respectively, and x and t are the position (with respect to the surface) and time (with respect to the step change), respectively. We will consider the case where $T_s=500$ K and $T_i=300$ K. The properties of the material include the thermal conductivity $k = 50$ W/m-K, density $\rho = 8000$ kg/m³, and $c = 400$ J/kg-K. The thermal diffusivity, α , is defined as:

$$\alpha = \frac{k}{\rho c} \quad (\text{A.2.6})$$

The erf function is an esoteric function that you may never have seen before. Therefore, to evaluate the temperature with this solution you would probably need to refer to tabulated values of the erf function. To work with the solution (i.e., to integrate or differentiate the result) you would likely need to find a good reference book that includes the rules necessary for these operations. This is not convenient; the use of Maple and EES together allows you to work with this solution quickly and easily.

We can enter the known information in EES. (See Appendix A.1 to become familiar with EES).

k=50 [W/m-K]	"thermal conductivity"
rho=8000 [kg/m^3]	"density"
c=400 [J/kg-K]	"specific heat capacity"
T_i=300 [K]	"initial temperature"
T_s=500 [K]	"surface temperature"
alpha=k/(rho*c)	"thermal diffusivity"

Next, enter the expression for the temperature distribution in Maple where it can be manipulated symbolically.

```
> T:=T_i+(T_s-T_i)*(1-erf(x/(2*sqrt(alpha*t))));
```

$$T := T_i + (T_s - T_i) \left(1 - \operatorname{erf} \left(\frac{x}{2\sqrt{\alpha t}} \right) \right)$$

The rate of heat transfer per unit area at any given position and time is given by:

$$\dot{q}'' = -k \frac{\partial T}{\partial x} \quad (\text{A.2.7})$$

which can be evaluated symbolically in Maple:

```
> q := -k*diff(T,x);
```

$$q := -\frac{k(-T_s + T_i) e^{\left(-\frac{x^2}{4\alpha t}\right)}}{\sqrt{\pi} \sqrt{\alpha t}}$$

The expression for the heat flux can be copied in Maple and pasted into EES (highlight the equation in Maple, select Copy from the Edit menu and then paste in the EES equation window):

```
q := -k*(-T_s+T_i)/Pi^(1/2)*exp(-1/4*x^2/alpha/t)/(alpha*t)^(1/2)
```

The equation must be manipulated slightly in order to be compatible with EES (in this case, the := must be changed to =):

```
q = -k*(-T_s+T_i)/Pi^(1/2)*exp(-1/4*x^2/alpha/t)/(alpha*t)^(1/2)
```

The expression can easily be evaluated, plotted, etc. in EES and the unit consistency of the input parameters and the expression can be checked. For example, Figure A.2-1 shows the heat flux as a function of position at various times; the figure was generated by setting the variable t in the Equation window and using a parametric table to evaluate the heat flux over a range of position.




Figure A.2-1: Heat flux as a function of position for various values of time.

The heat flux at the surface ($\dot{q}_{x=0}''$) is obtained by evaluating the heat flux at $x=0$, which can be done symbolically in Maple using the **eval** command.

```
> qs:=eval(q,x=0);
```

$$qs := -\frac{k(-T_s + T_i)}{\sqrt{\pi} \sqrt{\alpha t}}$$

and the energy per unit area that has been transferred to the solid (E'') is obtained by integrating the heat flux at the surface over time:

$$E'' = \int_0^t \dot{q}_{x=0}'' dt \quad (\text{A.2.8})$$

which can be obtained using the **int** command in Maple:

```
> E:=int(qs,t=0..t);
```

$$E := -\frac{2 t k (-T_s + T_i)}{\sqrt{\pi} \sqrt{\alpha t}}$$

The expression for E'' can be copied and pasted into EES (note that, again, the $:=$ must be changed to $=$):

```
E = -2*t*k*(-T_s+T_i)/Pi^(1/2)/(alpha*t)^(1/2)
```

in order to generate Figure A.2-2, which shows that amount of energy per unit area added to the solid as a function of time.




Figure A.2-2: Energy per unit area added to the solid as a function of time.

The details of the heat transfer problem are not important here; rather, it is important that you understand how to use Maple and EES together to solve engineering problems.

A.2.6 Solving Differential Equations

The study of heat transfer is in large part the study of solving differential equations that happen to involve temperature. It is not possible to make much progress without being able to solve a variety of differential equations that range from relatively simple 1st order to more complex 2nd order and even partial differential equations. The combination of Maple (again, to carry out the symbolic mathematical manipulations) and EES (to implement the symbolic solution in a useful way) empowers you to focus on the underlying concepts and manipulating the solution to carry out useful design studies rather than spending all of your time in the details of the math.

A.2.6.1 Entering Differential Equations

Differential equations are entered using the **diff** command together with a placeholder for the unknown solution expressed in terms of its independent variables. For example, the 1st order differential equation:

$$\rho V c \frac{dT}{dt} = -h A_s (T - T_f) \quad (\text{A.2.9})$$

would be entered in Maple as:

```
> restart;
> GDE:=rho*V*c*diff(T(t),t)=-h*As*(T(t)-T_f);
      GDE := ρ V c  $\left(\frac{d}{dt} T(t)\right) = -h A_s (T(t) - T_f)$ 
```

where GDE is a name for the equation provided by the user. (Here, GDE represents the governing differential equation.) The equation can be solved and the undetermined solution, T in Eq. (A.2.9), is indicated with $T(t)$ in order to signify that it is a function only of time.

The 2nd order differential equation:

$$\frac{d^2 T}{dx^2} = -\frac{\dot{g}''' \exp\left(-\frac{x}{L}\right)}{k} \quad (\text{A.2.10})$$

would be entered as:

```
> restart;
> GDE:=diff(diff(T(x),x),x)=-gdot*exp(-x/L)/k;
```

$$GDE := \frac{d^2}{dx^2} T(x) = -\frac{g \cdot e^{-\frac{x}{L}}}{k}$$

The partial differential equation:

$$\alpha \frac{\partial^2 T}{\partial x^2} = \frac{\partial T}{\partial t} \quad (\text{A.2.11})$$

would be entered as:

```
> restart;
> GDE:=alpha*diff(diff(T(x,t),x),x)=diff(T(x,t),t);
```

$$GDE := \alpha \left(\frac{\partial^2}{\partial x^2} T(x, t) \right) = \frac{\partial}{\partial t} T(x, t)$$

Notice that the temperature in Eq. (A.2.11) is a function of both x and t and therefore the function $T(x,t)$ is used in Maple to show that the solution is a function of both position and time.

A.2.6.2 Solving Differential Equations

Maple can obtain the solution to many of the differential equations that arise during the study of heat transfer. Once the differential equation has been entered, as discussed in Section A.2.6.1, the **dsolve** command is used to obtain the solution to within the undetermined constants.

For example, to obtain the solution to the 1st order differential equation, Eq. (A.2.9):

```
> restart;
> GDE:=rho*V*c*diff(T(t),t)=-h*As*(T(t)-T_f);
```

$$GDE := \rho V c \left(\frac{d}{dt} T(t) \right) = -h A_s (T(t) - T_f)$$

```
> Ts:=dsolve(GDE);
```

$$Ts := T(t) = T_f + e^{-\frac{h A_s t}{\rho V c}} _CI$$

where the variable $_CI$ in the solution is the undetermined constant of integration which must be established using the boundary condition. The solution to the 2nd order differential equation, Eq. (A.2.10), includes 2 constants of integration:

```
> restart;
> GDE:=diff(diff(T(x),x),x)=-gdot*exp(-x/L)/k;
```

$$GDE := \frac{d^2}{dx^2} T(x) = -\frac{g \cdot e^{-\frac{x}{L}}}{k}$$

```
> Ts:=dsolve(GDE);
```

$$Ts := T(x) = -\frac{L^2 \text{gdot} e^{\left(-\frac{x}{L}\right)}}{k} + _C1 x + _C2$$

The solutions to the differential equation can be cut and pasted directly into EES for evaluation and manipulation in the context of a specific problem (for example, $\text{g}''' = 5e5 \text{ W/m}^3$, $k = 10 \text{ W/m-K}$, and $L = 10 \text{ cm}$). The inputs are entered in EES:

```
gdot=1e5 [W/m^3]      "volumetric generation rate"
k=10 [W/m-K]         "conductivity"
L=10 [cm]*convert(cm,m) "length"
```

The solution is copied from Maple:

```
Ts := T(x) = -L^2*gdot*exp(-x/L)/k+_C1*x+_C2 "solution from Maple"
```

Note that some manipulation is required, specifically the $Ts :=$ must be removed, the (x) must be removed from T and the $_C1$ and $_C2$ are replaced with C_1 and C_2 .

```
T = -L^2*gdot*exp(-x/L)/k+C_1*x+C_2 "solution modified to be consistent with EES"
```

A.2.6.3 Evaluating Boundary Conditions

Note that there are 7 variables in the EES file but only 4 equations; we are three pieces of information short of obtaining a solution (C_1 , C_2 , and x). The boundary conditions must be used to establish the value of the undetermined constants. Maple can manipulate the solution in order to obtain symbolic expressions for the boundary conditions which can be entered in EES in order to compute these constants.

For example, the boundary conditions to Eq. (A.2.10) might be:

$$\left. \frac{dT}{dx} \right|_{x=0} = 0 \quad (\text{A.2.12})$$

and

$$-k \left. \frac{dT}{dx} \right|_{x=L} = \bar{h} (T_{x=L} - T_\infty) \quad (\text{A.2.13})$$

where $\bar{h} = 100 \text{ W/m}^2\text{-K}$ and $T_\infty = 100^\circ\text{C}$. The solution to the differential equation must be substituted into Eqs. (A.2.12) and (A.2.13) in order to obtain two symbolic expressions for the two constants. The solution (Ts in Maple) can be manipulated using the **diff**, **int**, and **eval** commands in order to accomplish this substitution. For example, to obtain a symbolic expression for the temperature gradient, use the **diff** command:

```
> dTdx:=diff(Ts,x);
```

$$\frac{d}{dx} T(x) = \frac{L \text{ gdot } e^{\left(-\frac{x}{L}\right)}}{k} + _C1$$

Note that it is possible to keep only the right-hand side of the result of the **diff** command using the **rhs** function; this is useful when you are trying to obtain a concise, symbolic statement of a boundary condition that can be pasted without much modification into EES.

```
> rhs(diff(Ts,x));
```

$$\frac{L \text{ gdot } e^{\left(-\frac{x}{L}\right)}}{k} + _C1$$

To obtain a symbolic expression for Eq. (A.2.12), use the **eval** command to evaluate the temperature gradient at $x = 0$:

```
> rhs(eval(diff(Ts,x),x=0))=0;
```

$$\frac{L \text{ gdot}}{k} + _C1 = 0$$

which can be copied and pasted into EES:

```
L*gdot/k+_C_1 = 0 "boundary condition #1"
```

and a similar process can be used to obtain a symbolic expression for Eq. (A.2.13):

```
> -k*rhs(eval(diff(Ts,x),x=0))=h_bar*(rhs(eval(Ts,x=L))-T_infinity);
```

$$-k \left(\frac{L \text{ gdot}}{k} + _C1 \right) = h_bar \left(-\frac{L^2 \text{ gdot } e^{(-1)}}{k} + _C1 L + _C2 - T_infinity \right)$$

which can also be copied and pasted into EES:

```
h_bar=100 [W/m^2-K] "average heat transfer coefficient"
T_infinity=converttemp(C,K,100 [C]) "ambient temperature"
-k*(L*gdot/k+_C_1) = h_bar*(-L^2*gdot*exp(-1)/k+_C_1*L+_C_2-T_infinity) "boundary condition #2"
```

Your EES code should be:

```
gdot=1e5 [W/m^3] "volumetric generation rate"
k=10 [W/m-K] "conductivity"
L=10 [cm]*convert(cm,m) "length"
T = -L^2*gdot*exp(-x/L)/k+_C_1*x+_C_2 "solution modified to be consistent with EES"
L*gdot/k+_C_1 = 0 "boundary condition #1"
```

```

h_bar=100 [W/m^2-K]           "average heat transfer coefficient"
T_infinity=converttemp(C,K,100 [C]) "ambient temperature"
-k*(L*gdot/k+C_1) = h_bar*(-L^2*gdot*exp(-1)/k+C_1*L+C_2-T_infinity) "boundary condition #2"

```

It is possible to set and check the units for each variable. A parametric table can be used to evaluate the temperature over a range of position and generate the plot shown in Figure A.2-3.




Figure A.2-3: Temperature as a function of position.

Reference:

Abell, M.L. and Braselton, J.P., *Maple V By Example*, 2nd edition, Academic Press, San Diego, CA, <http://www.apnet.com>, 1999.